Some examples of Linear Algebra in Maple

Vector input and arithmetic

```
> u := <1,1>;
```

$$u := \begin{bmatrix} 1 \\ 1 \end{bmatrix} \quad \textbf{(1)}$$

```
> v := Vector([2,1]);
```

$$v := \begin{bmatrix} 2 \\ 1 \end{bmatrix} \quad \textbf{(2)}$$

```
> u.v; # dot product
```

$$3 \quad \textbf{(3)}$$

```
> u+v;
```

$$\begin{bmatrix} 3 \\ 2 \end{bmatrix} \quad \textbf{(4)}$$

Matrix input and arithmetic

```
> A := Matrix([[2,1],[1,2]]);
```

$$A := \begin{bmatrix} 2 & 1 \\ 1 & 2 \end{bmatrix} \quad \textbf{(5)}$$

```
> B := Matrix([[1,1],[1,0]]);
```

$$B := \begin{bmatrix} 1 & 1 \\ 1 & 0 \end{bmatrix} \quad \textbf{(6)}$$

```
> A+B;
```

$$\begin{bmatrix} 3 & 2 \\ 2 & 2 \end{bmatrix} \quad \textbf{(7)}$$

```
> A.B;
```

$$\begin{bmatrix} 3 & 2 \\ 3 & 1 \end{bmatrix} \quad \textbf{(8)}$$

```
> A.u;
```

$$\begin{bmatrix} 3 \\ 3 \end{bmatrix} \quad \textbf{(9)}$$

```
> (A^2).u;
```

$$\begin{bmatrix} 9 \\ 9 \end{bmatrix} \quad \textbf{(10)}$$

```
> A.(A.u);
```

(11)

$$\begin{bmatrix} 9 \\ 9 \end{bmatrix} \quad (11)$$

Solving $A \cdot x = b$

```
> b := <1,1>;
```

$$b := \begin{bmatrix} 1 \\ 1 \end{bmatrix} \quad (12)$$

```
> Ab := <A|b>;
```

$$Ab := \begin{bmatrix} 2 & 1 & 1 \\ 1 & 2 & 1 \end{bmatrix} \quad (13)$$

```
> with(LinearAlgebra):
> R := ReducedRowEchelonForm(Ab);
```

$$R := \begin{bmatrix} 1 & 0 & \frac{1}{3} \\ 0 & 1 & \frac{1}{3} \end{bmatrix} \quad (14)$$

```
> R[1..2,3];
```

$$\begin{bmatrix} \frac{1}{3} \\ \frac{1}{3} \end{bmatrix} \quad (15)$$

```
> LinearSolve(A,b);
```

$$\begin{bmatrix} \frac{1}{3} \\ \frac{1}{3} \end{bmatrix} \quad (16)$$

Matrix inverse

```
> I2 := <<1,0>|<0,1>>;
```

$$I2 := \begin{bmatrix} 1 & 0 \\ 0 & 1 \end{bmatrix} \quad (17)$$

```
> B := <A|I2>;
```

$$B := \begin{bmatrix} 2 & 1 & 1 & 0 \\ 1 & 2 & 0 & 1 \end{bmatrix} \quad (18)$$

```
> R := ReducedRowEchelonForm(B);
```

(19)

$$R := \begin{bmatrix} 1 & 0 & \frac{2}{3} & -\frac{1}{3} \\ 0 & 1 & -\frac{1}{3} & \frac{2}{3} \end{bmatrix} \quad (19)$$

```
> R[1..2,3..4];
```

$$\begin{bmatrix} \frac{2}{3} & -\frac{1}{3} \\ -\frac{1}{3} & \frac{2}{3} \end{bmatrix} \quad (20)$$

```
> 1/A;
```

$$\begin{bmatrix} \frac{2}{3} & -\frac{1}{3} \\ -\frac{1}{3} & \frac{2}{3} \end{bmatrix} \quad (21)$$

Determinants and characteristic polynomials

```
> A;
```

$$\begin{bmatrix} 2 & 1 \\ 1 & 2 \end{bmatrix} \quad (22)$$

```
> Determinant(A);
```

$$3 \quad (23)$$

```
> CharacteristicPolynomial(A,x);
```

$$x^2 - 4x + 3 \quad (24)$$

```
> x*I2-A;
```

$$\begin{bmatrix} x-2 & -1 \\ -1 & x-2 \end{bmatrix} \quad (25)$$

```
> Determinant(x*I2-A);
```

$$x^2 - 4x + 3 \quad (26)$$

Matrix input using a loop

```
> T4 := Matrix(4,4);
```

$$T4 := \begin{bmatrix} 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \end{bmatrix} \quad (27)$$

```
> for i to 4 do
      for j to 4 do
          T4[i,j] := abs(i-j);
      od;
```

```
od;
T4;
```

$$\begin{bmatrix} 0 & 1 & 2 & 3 \\ 1 & 0 & 1 & 2 \\ 2 & 1 & 0 & 1 \\ 3 & 2 & 1 & 0 \end{bmatrix} \quad \textbf{(28)}$$

```
> C := [x[1],x[2],x[3],x[4]];
  for i to 4 do
      for j to 4 do
          T4[i,j] := C[ abs(i-j)+1 ];
      od;
  od;
  T4;
```

$$C := [x_1, x_2, x_3, x_4]$$

$$\begin{bmatrix} x_1 & x_2 & x_3 & x_4 \\ x_2 & x_1 & x_2 & x_3 \\ x_3 & x_2 & x_1 & x_2 \\ x_4 & x_3 & x_2 & x_1 \end{bmatrix} \quad \textbf{(29)}$$

```
> Determinant(T4);
```

$$x_1^4 - 3\,x_1^2\,x_2^2 - 2\,x_1^2\,x_3^2 - x_1^2\,x_4^2 + 4\,x_1\,x_2^2\,x_3 + 4\,x_1\,x_2\,x_3\,x_4 + x_2^4 - 2\,x_2^3\,x_4 - 2\,x_2^2\,x_3^2 + x_2^2\,x_4^2 - 2\,x_2\,x_3^2\,x_4 + x_3^4 \quad \textbf{(30)}$$